# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Maintainer: Charles Mauch <cmauch@gmail.com>

pkgname=perl-app-cli
pkgver=0.52
pkgrel=4
pkgdesc="Perl/CPAN Module App::CLI"
arch=("any")
url="https://search.cpan.org/dist/App-CLI"
license=("GPL" "PerlArtistic")
depends=("perl")
source=("https://cpan.metacpan.org/authors/id/P/PT/PTC/App-CLI-$pkgver.tar.gz")
sha256sums=('52bd43f555913cc2ffd6405f9951d2aabd46af63d701d9b5e346a6332709f0ce')

build() {
  cd "$srcdir"/App-CLI-$pkgver
  PERL_USE_UNSAFE_INC=1 \
  PERL_MM_USE_DEFAULT=1 perl Makefile.PL INSTALLDIRS=vendor
  make
}

package() {
  cd "$srcdir"/App-CLI-$pkgver
  make pure_install doc_install DESTDIR="$pkgdir"
  find "$pkgdir" -name '.packlist' -delete
  find "$pkgdir" -name '*.pod' -delete
}
